




Data: Reading, Analyzing, Plotting



Oregon State
University

Mike Bailey

mjb@cs.oregonstate.edu



Oregon State
University
Computer Graphics

Reading from a File, I

Everything is done in `setup()` because it only needs to happen once

```
void
setup( )
{
    size( 800, 800 );
    noFill( );

    String [ ] lines = loadStrings( "data.txt" );
    if( lines == null )
    {
        println( "Cannot open data.txt" );
        exit( );
    }

    int numPoints = int( lines[0] );
    println( "numPoints = " + numPoints );
}
```



Reading from a File, II

Everything is done in setup() because it only needs to happen once

```
float [ ] x = new float [ numPoints ];
float [ ] y = new float [ numPoints ];
for( int i = 0; i < numPoints; i = i + 1 )
{
    y[ i ] = int( lines[i+1] );
    println( "y[" + i + "] = " + y[ i ] );
}
```



Reading from a File, III

Everything is done in setup() because it only needs to happen once

```
float sum = 0.;
for( int i = 0; i < numPoints; i = i + 1 )
{
    sum = sum + y[ i ];
}
float average = sum / float(numPoints);
println( "average = " + average );

sum = 0.;
for( int i = 0; i < numPoints; i = i + 1 )
{
    float diff = y[ i ] - average;
    sum = sum + ( diff * diff );
}
float stdev = sqrt( sum / float(numPoints - 1) );
println( "stdev = " + stdev );
```



Reading from a File, IV

Everything is done in setup() because it only needs to happen once

```

float ymin = y[ 0 ];
float ymax = y[ 0 ];
for( int i = 1; i < numPoints; i = i + 1 )
{
  if( y[ i ] < ymin )
    ymin = y[ i ];
  if( y[ i ] > ymax )
    ymax = y[ i ];
}

float xscale = float(width) / float(numPoints - 1);
float yscale = float(height) / ( ymax - ymin );

background( 200, 255, 200 );
stroke( 0, 0, 255 );
strokeWeight( 3 );

beginShape( );
for( int i = 0; i < numPoints; i = i + 1 )
{
  vertex( xscale * float( i ), height - yscale * ( y[ i ] - ymin ) );
}
endShape( );
}

```

Find the minimum and maximum values so we know how to scale the vertical part of the graph




Oregon State
University

Computer Graphics

Reading from a File, V

The Data File:

12 ← Number of Points
47.
51.
56.
61.
67.
73.
82. ← Average monthly temperatures in Corvallis
83.
77.
65.
53.
46.



Challenge question:
How could you draw little circles at each data point?